Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
NotebookOpen[wdir <> "\\MakeVSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization / Utilities

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 2; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 1; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
  Collect[Normal@Series[ℰ /. {T_i_ → e^(ℏ t_i/2), T → e^(ℏ t/2)}, {ℏ, 0, $TℏD}], ℏ, Together]]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

Differential polynomials (DP):

Utils

```
DP_{α_→D_x_, β_→D_y_}[P_][λ_] :=
 Total[CoefficientRules[P, {α, β}] /. ({m_, n_} → c_) ↦ c D[λ, {x, m}, {y, n}]]
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, CE, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
   },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  CE[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  U_i_[NCM[]] := U[];
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := CE[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, CE[a b (x ** y)]];
  (a_ * x_U) ** y_ := CE[a (x ** y)]; x_ ** (a_ * y_U) := CE[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ l_null, {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. l_s_ ⧴ (l /. x_i_ ⧴ x_s));
    CE[Total[
       CoefficientRules[poly, vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
   ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := CE@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) ⧴ c NCM @@ MapThread[pow, {Last /@ {ss}, p}]];
  S_i_[c_. * u_U] := CE[(c /. S_i[U, Centrals]) DeleteCases[u, _i] **
     U_i[NCM @@ Reverse@Cases[u, x_i ⧴ S@U@x]]];
 ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) ⧴ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U ⧴ m[u]];)
```

Meta-Operations

QLImplementation

```
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
(S@CU@y = -CU@y; S@CU@a = -CU@a; S@CU@x = -CU@x;)
S_i_[CU, Centrals] = {t_i → -t_i};
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas}]]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{0.703125, {(28 t² γ⁴ + 116 t γ⁵ ϵ + 120 γ⁶ ϵ²) CU[y, y, y, x, x] +
    <<23>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

```
S₁[CU[y₁, a₁, x₁]]
```

```
-2 ϵ CU[a₁, a₁] + 2 γ CU[y₁, x₁] - CU[y₁, a₁, x₁] - γ CU[] t₁ + CU[a₁] (2 γ ϵ + t₁)
```

Verifying that S is an anti-homomorphism on CU:

```
With[{bas = CU /@ {y₁, a₁, x₁}},
 Table[HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas}] ]
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying the involutivity of S on products of triples:

```
With[{bas = CU /@ {y₁, a₁, x₁}},
 Table[HL@Simp[z1 ** z2 ** z3 - S₁@S₁[z1 ** z2 ** z3]],
  {z1, bas}, {z2, bas}, {z3, bas}] ]
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Representing $sl_2^{\gamma\epsilon}$

$\rho t = \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$; $\rho x = \begin{pmatrix} 0 & 1 \\ 0 & 0 \end{pmatrix}$; $\rho a = \frac{1}{2}\begin{pmatrix} \gamma + 1/\epsilon & 0 \\ 0 & -\gamma + 1/\epsilon \end{pmatrix}$; $\rho y = \begin{pmatrix} 0 & 0 \\ \epsilon\gamma & 0 \end{pmatrix}$;

```
{ρa.ρx - ρx.ρa == γ ρx, ρa.ρy - ρy.ρa == -γ ρy, ρx.ρy - ρy.ρx == 2 ϵ ρa - ρt} // Simplify
{True, True, True}
```

Crho

$\{C\rho@y, C\rho@a, C\rho@x\} = \left\{\begin{pmatrix} 0 & 0 \\ \epsilon & 0 \end{pmatrix}, \begin{pmatrix} \gamma & 0 \\ 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & \gamma \\ 0 & 0 \end{pmatrix}\right\};$

$C\rho[\mathcal{E}\_] := \mathcal{E} /. \{t \to \gamma\,\epsilon, CU[u\_\_\_] :\to Dot[\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, Sequence @@ (C\rho /@ \{u\})]\}$

Verifying that Cρ represents CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cρ[z1 ** z2] == Cρ[z1].Cρ[z2] // Simplify // HL,
  {z1, bas}, {z2, bas}] ]
{{True, True, True}, {True, True, True}, {True, True, True}}
```

```
MatrixForm /@ {Cρ@y, Cρ@a, Cρ@x}
```

$\left\{\begin{pmatrix} 0 & 0 \\ \epsilon & 0 \end{pmatrix}, \begin{pmatrix} \gamma & 0 \\ 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & \gamma \\ 0 & 0 \end{pmatrix}\right\}$

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, $T = e^{\hbar t/2}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle / \mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right)/\hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[e^(γ ϵ ℏ)]; (*T=SS[e^(ℏ t/2)];*)
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] = (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T^2 e^(-2 ϵ a ℏ)) / ℏ], {a}];
(S@QU@y = 𝕆_QU[SS[-T^-2 e^(ℏ ϵ a) y], {a, y}];
 S@QU@a = -QU@a; S@QU@x = 𝕆_QU[SS[-e^(ℏ ϵ a) x], {a, x}];)
S_i_[QU, Centrals] = {t_i → -t_i, T_i → T_i^-1};
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$$\Big\{\Big\{\{QU[y],QU[y]\}\to 0,\ \{QU[y],QU[a]\}\to\gamma\,QU[y],\ \{QU[y],QU[x]\}\to \left(t+\frac{t^2\hbar}{2}+\frac{t^3\hbar^2}{6}+\frac{t^4\hbar^3}{24}+\frac{t^5\hbar^4}{120}\right)QU[\,]+\left(-2\epsilon-2t\epsilon\hbar-t^2\epsilon\hbar^2-\frac{1}{3}t^3\epsilon\hbar^3-\frac{1}{12}t^4\epsilon\hbar^4\right)QU[a]+\left(2\epsilon^2\hbar+2t\epsilon^2\hbar^2+t^2\epsilon^2\hbar^3+\frac{1}{3}t^3\epsilon^2\hbar^4\right)QU[a,a]+\left(-\gamma\epsilon\hbar-\frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)QU[y,x]\Big\},$$

$$\{\{QU[a],QU[y]\}\to-\gamma\,QU[y],\ \{QU[a],QU[a]\}\to 0,\ \{QU[a],QU[x]\}\to\gamma\,QU[x]\},$$

$$\Big\{\{QU[x],QU[y]\}\to \left(-t-\frac{t^2\hbar}{2}-\frac{t^3\hbar^2}{6}-\frac{t^4\hbar^3}{24}-\frac{t^5\hbar^4}{120}\right)QU[\,]+\left(2\epsilon+2t\epsilon\hbar+t^2\epsilon\hbar^2+\frac{1}{3}t^3\epsilon\hbar^3+\frac{1}{12}t^4\epsilon\hbar^4\right)QU[a]+\left(-2\epsilon^2\hbar-2t\epsilon^2\hbar^2-t^2\epsilon^2\hbar^3-\frac{1}{3}t^3\epsilon^2\hbar^4\right)QU[a,a]+\left(\gamma\epsilon\hbar+\frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)QU[y,x],$$

$$\{QU[x],QU[a]\}\to-\gamma\,QU[x],\ \{QU[x],QU[x]\}\to 0\Big\}\Big\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}

Verifying associativity on a “random” triple (~34 secs @ $TℏD=5, $TϵD=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp ]
  }] // Timing
```

$$\Big\{66.7031,\ \Big\{\left(28t^2\gamma^4+116t\gamma^5\epsilon+120\gamma^6\epsilon^2+\ll 10\gg+\frac{757}{15}t^5\gamma^5\epsilon\hbar^4+\frac{1785}{4}t^4\gamma^6\epsilon^2\hbar^4\right)QU[y,y,y,x,x]+\ll 21\gg+\ll 1\gg,\ 0\Big\}\Big\}$$

Verifying that *S* is an anti-homomorphism on QU:

```
With[{bas = QU /@ {y₁, a₁, x₁}},
 Table[{z1, z2} → HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas}]]
```

```
{{{QU[y₁], QU[y₁]} → 0, {QU[y₁], QU[a₁]} → 0, {QU[y₁], QU[x₁]} → 0},
 {{QU[a₁], QU[y₁]} → 0, {QU[a₁], QU[a₁]} → 0, {QU[a₁], QU[x₁]} → 0},
 {{QU[x₁], QU[y₁]} → 0, {QU[x₁], QU[a₁]} → 0, {QU[x₁], QU[x₁]} → 0}}
```

Verifying that $\mathrm{Lim}_{\hbar\to 0}$ QU = CU using a "random" product (~23 secs @ \$T$\hbar$D=5, \$TϵD=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → e^(ħ t/2)}, ħ → 0] - lhs] // HL
  }] // Timing
```

$$\{51.0313,\ \{2\,(8\,t^2\,\gamma^4 + 16\,t\,\gamma^5\,\epsilon)\,\mathrm{CU}[y,y,y,x,x] + \ll 107 \gg + \mathrm{CU}[y,y,y,y,y,a,a,a,a,x,x,x,x],$$
$$2\left(8\,t\,\gamma^6\,\epsilon^2\,\hbar + 12\,t^2\,\gamma^6\,\epsilon^2\,\hbar^2 + \frac{28}{3}\,t^3\,\gamma^6\,\epsilon^2\,\hbar^3 + 5\,t^4\,\gamma^6\,\epsilon^2\,\hbar^4\right)\mathrm{QU}[y,y,y,x,x] +$$
$$\ll 566 \gg + (\ll 1 \gg)\ \ll 1 \gg,\ 0\}\}$$

Representing $\mathcal{U}_{\gamma\epsilon;\hbar}$

Qrho

$$\{\mathrm{Q}\rho@y,\ \mathrm{Q}\rho@a,\ \mathrm{Q}\rho@x\} = \left\{\begin{pmatrix}0 & 0\\ \epsilon & 0\end{pmatrix}, \begin{pmatrix}\gamma & 0\\ 0 & 0\end{pmatrix}, \begin{pmatrix}0 & \mathrm{SS}@\frac{1-e^{-\gamma\epsilon\hbar}}{\epsilon\hbar}\\ 0 & 0\end{pmatrix}\right\};$$
$$\mathrm{Q}\rho[\mathcal{E}\_] := \mathcal{E}\ /.\ \{t \to \gamma\,\epsilon,\ \mathrm{QU}[u\_\_\_] :\!\to \mathrm{Dot}[\begin{pmatrix}1 & 0\\ 0 & 1\end{pmatrix},\ \mathrm{Sequence} @@ (\mathrm{Q}\rho\ /@\ \{u\})]\}$$

Verifying that Cρ represents CU:

```
With[{bas = QU /@ {y, a, x}},
 Table[Qρ[z1 ** z2] == Qρ[z1].Qρ[z2] // Simplify // HL,
  {z1, bas}, {z2, bas}]]
```

```
{{True, True, True}, {True, True, True}, {True, True, True}}
```

```
MatrixForm /@ {Qρ@y, Qρ@a, Qρ@x}
```

$$\left\{\begin{pmatrix}0 & 0\\ \epsilon & 0\end{pmatrix}, \begin{pmatrix}\gamma & 0\\ 0 & 0\end{pmatrix}, \begin{pmatrix}0 & \gamma - \frac{1}{2}\gamma^2\,\epsilon\,\hbar + \frac{1}{6}\gamma^3\,\epsilon^2\,\hbar^2\\ 0 & 0\end{pmatrix}\right\}$$

Implementing θ

theta

$$\mathrm{DeclareMorphism}[\mathrm{C}\theta,\ \mathrm{CU} \to \mathrm{CU},\ \{y \to -\mathrm{CU}@x,\ a \to -\mathrm{CU}@a,\ x \to -\mathrm{CU}@y\},\ \{t \to -t,\ T \to T^{-1}\}];$$
$$\mathrm{DeclareMorphism}[\mathrm{Q}\theta,\ \mathrm{QU} \to \mathrm{QU},\ \{y \to \mathbb{O}_{\mathrm{QU}}[\mathrm{SS}[-T^{-1}\,e^{\hbar\epsilon a}\,x],\ \{a, x\}],$$
$$a \to -\mathrm{QU}@a,\ x \to \mathbb{O}_{\mathrm{QU}}[\mathrm{SS}[-T^{-1}\,e^{\hbar\epsilon a}\,y],\ \{a, y\}]\},\ \{t \to -t,\ T \to T^{-1}\}]$$

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}]]
```

{CU[y] → −CU[x] → CU[y], CU[a] → −CU[a] → CU[a], CU[x] → −CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}]]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}]]
```

$\{QU[y] \to \left(-1+\frac{t\hbar}{2}-\frac{t^2\hbar^2}{8}+\frac{t^3\hbar^3}{48}-\frac{t^4\hbar^4}{384}\right) QU[x] + \left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3+\frac{1}{48}t^3\epsilon\hbar^4\right) QU[a,x] + \left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3-\frac{1}{16}t^2\epsilon^2\hbar^4\right) QU[a,a,x] \to QU[y],$

$QU[a] \to -QU[a] \to QU[a],\ QU[x] \to \left(-1+\frac{t\hbar}{2}+\gamma\epsilon\hbar-\frac{t^2\hbar^2}{8}-\frac{1}{2}t\gamma\epsilon\hbar^2-\frac{1}{2}\gamma^2\epsilon^2\hbar^2+\frac{t^3\hbar^3}{48}+\frac{1}{8}t^2\gamma\epsilon\hbar^3+\frac{1}{4}t\gamma^2\epsilon^2\hbar^3-\frac{t^4\hbar^4}{384}-\frac{1}{48}t^3\gamma\epsilon\hbar^4-\frac{1}{16}t^2\gamma^2\epsilon^2\hbar^4\right) QU[y] + \left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2+\gamma\epsilon^2\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3-\frac{1}{2}t\gamma\epsilon^2\hbar^3+\frac{1}{48}t^3\epsilon\hbar^4+\frac{1}{8}t^2\gamma\epsilon^2\hbar^4\right) QU[y,a] + \left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3-\frac{1}{16}t^2\epsilon^2\hbar^4\right) QU[y,a,a] \to QU[x]\}$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}]]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\mathbb{AD}\$f = \frac{\gamma}{\hbar} e^{\hbar\left(\frac{t}{2}-(a+\gamma)\epsilon\right)} \left(\left(\mathrm{Cosh}\left[\hbar\left(a\epsilon+\frac{\gamma\epsilon}{2}-\frac{t}{2}\right)\right]-\mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\gamma\epsilon}{2}\right)^2+\epsilon\omega}\right]\right)\Big/\left(\mathrm{Sinh}\left[\frac{\gamma\epsilon\hbar}{2}\right]\left(a^2\epsilon+a\gamma\epsilon-a t-\omega\right)\right)\right);$$

```
SS[A𝔻$f] // Simplify
```

$$\frac{1}{720}\left(720+360\left(t-2\,(a+\gamma)\,\epsilon\right)\hbar+60\left(2\,t^2-7\,t\,(a+\gamma)\,\epsilon+\epsilon\left(7\,a^2\,\epsilon+13\,a\,\gamma\,\epsilon+6\,\gamma^2\,\epsilon+\omega\right)\right)\hbar^2+30\left(t^3-5\,t^2\,(a+\gamma)\,\epsilon-2\,(a+\gamma)\,\epsilon^2\,\omega+t\,\epsilon\left(9\,a^2\,\epsilon+17\,a\,\gamma\,\epsilon+8\,\gamma^2\,\epsilon+\omega\right)\right)\hbar^3+\left(6\,t^4-39\,t^3\,(a+\gamma)\,\epsilon-t\left(32\,a+33\,\gamma\right)\epsilon^2\,\omega+2\,\epsilon^2\,\omega^2+t^2\,\epsilon\left(101\,a^2\,\epsilon+192\,a\,\gamma\,\epsilon+91\,\gamma^2\,\epsilon+9\,\omega\right)\right)\hbar^4\right)$$

Scaling behaviour of A𝔻$f:

```
Simplify[A𝔻$f == ((A𝔻$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
```

```
True
```

```
FullSimplify[
 A𝔻$f == ((A𝔻$f /. γ → 1) /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

```
True
```

ADeq

```
A𝔻$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[A𝔻, QU → CU,
 {a → CU@a, x → CU@x, y → 𝕊_CU[SS[A𝔻$f], a → CU[a], ω → A𝔻$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[A𝔻[z1 ** z2] - A𝔻[z1] ** A𝔻[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

$$\texttt{S𝔻\$g} = \sqrt{\left(\left(\mathrm{Cosh}\left[\frac{\hbar}{2}\sqrt{t^2+\gamma^2\,\epsilon^2+4\,\epsilon\,\varpi}\right]-\mathrm{Cosh}\left[\frac{\hbar}{2}\left(t-\left(2\,a+\gamma\right)\epsilon\right)\right]\right)\Big/\left(\mathrm{Sinh}\left[\frac{\gamma\,\epsilon\,\hbar}{2}\right]\left(t\left(2\,a+\gamma\right)-2\,a\,(a+\gamma)\,\epsilon+2\,\varpi\right)\hbar\,/\,(2\,\gamma)\right)\right)};$$

```
{SD$P = (Cosh[ħ ((ϵ-t)/2 + ϵ a)] - Cosh[ħ Sqrt[(t^2+ϵ^2)/4 + ϵ ϖ]]) /
        (ħ Sinh[-ϵ ħ/2] (ϖ - ϵ a^2 + (t - ϵ) a + t/2)),
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ħ → γ^2 ħ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
    SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 Simplify[SD$g == FullSimplify[
    Sqrt[SD$Q] /. c → a + 1/2 /. {ħ → γ^2 ħ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\left\{-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}\,(-t+\epsilon)\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}\,(t^2+\epsilon^2)+\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\,(t-\epsilon)-a^2\,\epsilon+\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True},\right.$$

$$\left.-\left(\left(4\left(\mathrm{Cosh}\left[\frac{1}{2}\,(t-2\,c\,\epsilon)\,\hbar\right]-\mathrm{Cosh}\left[\frac{1}{2}\,\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\,\epsilon+4\,\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True}\right\}$$

```
FullSimplify[SD$g /. {ħ → γ^2 ħ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]
```

$$\sqrt{2}\,\sqrt{\left(\left(\gamma^2\left(-\mathrm{Cosh}\left[\frac{1}{2}\,\left(t-\left(2\,a+\gamma^2\right)\epsilon\right)\hbar\right]+\mathrm{Cosh}\left[\frac{1}{2}\,\sqrt{t^2+\gamma^4\,\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{1}{2}\,\gamma^2\,\epsilon\,\hbar\right]\right)\Big/\left(\left(t\,\left(2\,a+\gamma^2\right)-2\,a\,\left(a+\gamma^2\right)\epsilon+2\,\varpi\right)\hbar\right)\right)}$$

```
SS[SD$g]
```

$$1+\frac{1}{48}\left(t^2-2\,a\,t\,\epsilon-t\,\gamma\,\epsilon+2\,a^2\,\epsilon^2+2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^2+\frac{1}{23\,040}\left(t^4-4\,a\,t^3\,\epsilon-2\,t^3\,\gamma\,\epsilon+16\,a^2\,t^2\,\epsilon^2+16\,a\,t^2\,\gamma\,\epsilon^2-5\,t^2\,\gamma^2\,\epsilon^2+4\,t^2\,\epsilon\,\varpi+8\,a\,t\,\epsilon^2\,\varpi+4\,t\,\gamma\,\epsilon^2\,\varpi+12\,\epsilon^2\,\varpi^2\right)\hbar^4$$

SDeq

```
SD$f = FullSimplify[e^(ħ (t/2-ϵ a)) (SD$g /. {a → -a, t → -t})];
```

```
SS[SD$f]
```

$$1+\frac{1}{2}(t-2a\epsilon)\hbar+\frac{1}{48}\left(7t^2-26at\epsilon+t\gamma\epsilon+26a^2\epsilon^2-2a\gamma\epsilon^2+2\epsilon\varpi\right)\hbar^2+\frac{1}{96}(t-2a\epsilon)\left(3t^2-10at\epsilon+t\gamma\epsilon+10a^2\epsilon^2-2a\gamma\epsilon^2+2\epsilon\varpi\right)\hbar^3+\frac{1}{23040}\left(121t^4-844at^3\epsilon+62t^3\gamma\epsilon+2296a^2t^2\epsilon^2-376at^2\gamma\epsilon^2-5t^2\gamma^2\epsilon^2+124t^2\epsilon\varpi-472at\epsilon^2\varpi-4t\gamma\epsilon^2\varpi+12\epsilon^2\varpi^2\right)\hbar^4$$

SDeq

```
SD$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → CU@a,
   x → S_CU[SS[SD$f], a → CU[a], ϖ → SD$ϖ] ** CU@x,
   y → S_CU[SS[SD$g] , a → CU[a], ϖ → SD$ϖ] ** CU@y
  }]
```

Verifying the θ-symmetry:

```
Table[HL@Simplify[Cθ[SD[z]] == SD[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

{True, True, True}

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@Simp[SD[z1 ** z2] - SD[z1] ** SD[z2]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

R in QU.

Quesne's formula:

Quesne

$$\mathbb{e}_{q\_,n\_}[x\_] := \mathrm{Exp}\left[\sum_{k=1}^{n}\frac{(1-q)^k}{k\,(1-q^k)}x^k\right];\quad \mathbb{e}_{q\_}[x\_] := \mathbb{e}_{q,\$TeD}[x]$$

```
Table[Together@SeriesCoefficient[e_{ρ,5}[x], {x, 0, n}], {n, 0, 5}]
```

$$\left\{1,1,\frac{1}{1+\rho},\frac{1}{(1+\rho)(1+\rho+\rho^2)},\frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)},1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[e_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
```

{1, 1, 1, 1, 1, 1}

R

```
QU[R_{i_,j_}] := 𝕆_QU[SS[e^{ħ b₁ a₂} e_q[ħ y₁ x₂] /. b₁ → γ^{-1} (ϵ a₁ - t_i)], {y₁, a₁}_i, {a₂, x₂}_j];
QU[R^{-1}_{i_,j_}] := S_j@QU[R_{i,j}];
```

```
QU[R_{3,4}] // Short
```

$$QU[\,] + \frac{\epsilon\,\hbar\,QU[a_3, a_4]}{\gamma} + \ll 39 \gg + QU[y_3, y_3, a_4, x_4, x_4]\left(-\frac{\hbar^3 t_3}{2\gamma} + \frac{1}{4}\epsilon\,\hbar^4 t_3\right)$$

Verifying R2 (~2 secs @ $TħD=4, $TϵD=2):

```
QU[R_{1,2} ** R^{-1}_{1,2}] // Simp // HL // Timing
```

{2.95313, QU[]}

Verifying R3 (~156 secs @ $TħD=4, $TϵD=2):

```
{Short[lhs = QU[R_{1,2} ** R_{1,3} ** R_{2,3}]], HL@Simp[lhs - QU[R_{2,3} ** R_{1,3} ** R_{1,2}]]} // Timing
```

$$\left\{230.938, \left\{QU[\,] + \frac{\epsilon\,\hbar\,QU[a_1, a_2]}{\gamma} + \ll 1240 \gg + QU[y_1, a_3, x_3]\left(\frac{\hbar^3 t_2^2}{\gamma} + \frac{\hbar^4 t_2^3}{2\gamma}\right), 0\right\}\right\}$$

C𝕆 and Swaps

Swaps from Pensieve://Talks/Toulouse-1705/DogmaDemo.nb and from Pensieve://Talks/Sydney-1708/ExtraDetails@@.nb.

CdsO

```
CU@C𝕆[𝔼[ω_, L_, Q_, P_], specs___] := 𝕆_CU[SS[ω e^{ħ (L+Q)} P], specs]
```

```
CU@C𝕆[𝔼[1, t₁ a₂, t₁^{-1} (e^{t₁} - 1) y₁ x₂, 1 + ϵ x₁ y₂], {y₁, x₁}₁, {x₂, a₂, y₂}₂] // Short
```

$$CU[\,] + \ll 42 \gg + CU[y_1, x_1]\left(-\gamma\,\epsilon\,\hbar^2 t_2 + e^{t_1}\gamma\,\epsilon\,\hbar^2 t_2 + \ll 8 \gg + \frac{1}{6} e^{t_1}\gamma^3\epsilon\,\hbar^4 t_1^2 t_2\right)$$

```
HL[MatrixExp[ξ Cρ[CU[x]]].MatrixExp[α Cρ[CU[a]]] ==
  MatrixExp[α Cρ[CU[a]]].MatrixExp[e^{-γ α} ξ Cρ[CU[x]]]]
```

True

SW

```
SW_{x_{i_},a_{j_}}[C𝕆[𝔼[ω_, L_, Q_, P_], specs___]] := C𝕆[
  With[{q = e^{-γ α} ξ x_i + α a_j},
   𝔼[ω, L, ħ^{-1} e^{-γ α} ξ x_i + (Q /. x_i → 0), e^{-q} DP_{x_i→D_ξ, a_j→D_α}[P][e^q]] /. {α → ħ ∂_{a_j} L, ξ → ħ ∂_{x_i} Q}
  ], Sequence @@ ({specs} /. {l___, x_i, a_j, r___} :→ {l, a_j, x_i, r})]
```

```
c𝕆 = C𝕆[𝔼[1, t₁ a₂, t₁^{-1} (e^{t₁} - 1) y₁ x₂, 1 + ϵ x₁ y₂], {y₁, x₁}₁, {x₂, a₂, y₂}₂]
```

$$C\mathbb{O}\left[\mathbb{E}\left[1, a_2 t_1, \frac{(-1 + e^{t_1})\, x_2 y_1}{t_1}, 1 + \epsilon\, x_1 y_2\right], \{y_1, x_1\}_1, \{x_2, a_2, y_2\}_2\right]$$

```
SW_{x2,a2}[cω]
```

$$\mathbb{C}\mathbb{O}\left[\mathbb{E}\left[1,\ a_2\,t_1,\ \frac{e^{-\gamma\hbar t_1}\left(-1+e^{t_1}\right)x_2\,y_1}{t_1},\ 1+\epsilon\,x_1\,y_2\right],\ \{y_1,x_1\}_1,\ \{a_2,x_2,y_2\}_2\right]$$

```
With[{cω = CO[E[1, t1 a2, t1^-1 (e^t1 - 1) y1 x2, 1 + ϵ x1 y2], {y1, x1}1, {x2, a2, y2}2]},
 HL[CU[cω] == CU[cω // SW_{x2,a2}]]]
```

True

```
With[{cω = CO[E[ω, l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2, γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2,
      1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)], {y1, a1, x1}1, {x2, a2, y2}2]},
 {CU[cω] // Short, HL[CU[cω] == CU[cω // SW_{x2,a2}]]}
]
```

$$\left\{\mathrm{CU}[a_1,a_1,a_1]\left(\frac{1}{2}\epsilon\,\omega\,\hbar^2\,l_1\,l_{11}^2\,t_1^2+\epsilon\,\omega\,\hbar^2\,l_1\,l_{11}\,l_{21}\,t_1\,t_2+\frac{1}{2}\epsilon\,\omega\,\hbar^2\,l_1\,l_{21}^2\,t_2^2\right)+\ll 75\gg+\mathrm{CU}[\,]\left(\ll 1\gg\right),\right.$$

True}

Stitching Direct

```
MatrixExp[η1 Cρ[CU@y]].MatrixExp[α1 Cρ[CU@a]].
   MatrixExp[ξ1 Cρ[CU@x]].MatrixExp[η2 Cρ[CU@y]].MatrixExp[α2 Cρ[CU@a]].
   MatrixExp[ξ2 Cρ[CU@x]] // Simplify // MatrixForm
```

$$\begin{pmatrix} e^{\gamma(\alpha_1+\alpha_2)}\left(1+\gamma\,\epsilon\,\eta_2\,\xi_1\right) & e^{\gamma\alpha_1}\gamma\left(e^{\gamma\alpha_2}\xi_2+\xi_1\left(1+e^{\gamma\alpha_2}\gamma\,\epsilon\,\eta_2\,\xi_2\right)\right) \\ e^{\gamma\alpha_2}\epsilon\left(\eta_2+e^{\gamma\alpha_1}\eta_1\left(1+\gamma\,\epsilon\,\eta_2\,\xi_1\right)\right) & 1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1+e^{\gamma\alpha_2}\gamma\,\epsilon\left(\eta_2+e^{\gamma\alpha_1}\eta_1\left(1+\gamma\,\epsilon\,\eta_2\,\xi_1\right)\right)\xi_2 \end{pmatrix}$$

```
eqn = MatrixExp[η1 Cρ[CU@y]].MatrixExp[α1 Cρ[CU@a]].MatrixExp[ξ1 Cρ[CU@x]].
   MatrixExp[η2 Cρ[CU@y]].MatrixExp[α2 Cρ[CU@a]].MatrixExp[ξ2 Cρ[CU@x]] ==
  e^{τ0 ϵ γ} MatrixExp[η0 Cρ[CU@y]].MatrixExp[α0 Cρ[CU@a]].MatrixExp[ξ0 Cρ[CU@x]]
```

$$\left\{\left\{e^{\gamma\alpha_2}\left(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_2\,\xi_1\right),\ e^{\gamma\alpha_1}\gamma\,\xi_1+e^{\gamma\alpha_2}\gamma\left(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_2\,\xi_1\right)\xi_2\right\},\right.$$
$$\left\{e^{\gamma\alpha_2}\left(e^{\gamma\alpha_1}\epsilon\,\eta_1+\epsilon\,\eta_2\left(1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1\right)\right),\right.$$
$$\left.\left.1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1+e^{\gamma\alpha_2}\gamma\left(e^{\gamma\alpha_1}\epsilon\,\eta_1+\epsilon\,\eta_2\left(1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1\right)\right)\xi_2\right\}\right\}==$$
$$\left\{\left\{e^{\alpha0\,\gamma+\gamma\,\epsilon\,\tau0},\ e^{\alpha0\,\gamma+\gamma\,\epsilon\,\tau0}\gamma\,\xi0\right\},\left\{e^{\alpha0\,\gamma+\gamma\,\epsilon\,\tau0}\epsilon\,\eta0,\ e^{\gamma\,\epsilon\,\tau0}\left(1+e^{\alpha0\,\gamma}\gamma\,\epsilon\,\eta0\,\xi0\right)\right\}\right\}$$

```
sol = Block[{ϵ}, Solve[Thread[Flatten /@ eqn], {τ0, η0, α0, ξ0}]][[1]]
```

Solve: Inconsistent or redundant transcendental equation. After reduction, the bad equation is $\mathrm{Log}[e^{\gamma(\alpha0+\epsilon\,\tau0)}] - \mathrm{Log}[e^{\gamma\,\alpha_2}\,(e^{\gamma\,\mathrm{Subscript}[\ll 2\gg]} + e^{\mathrm{Times}[\ll 2\gg]}\,\gamma\,\epsilon\,\eta_2\,\xi_1)] == 0$.

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Solve: Equations may not give solutions for all "solve" variables.

$$\Big\{\tau0 \to \frac{1}{\gamma\,\epsilon}\Big(-\mathrm{Log}\big[e^{\alpha0\,\gamma}\big] + \mathrm{Log}\big[e^{\gamma\,\alpha_1+\gamma\,\alpha_2} + e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2\,\xi_1\big]\Big),$$

$$\eta0 \to \Bigg(e^{-\gamma\,\alpha_1}\Bigg(\frac{1}{2} + \frac{1}{2}e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_1\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_2\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma^2\,\epsilon^2\,\eta_1\,\eta_2\,\xi_1\,\xi_2 - \frac{1}{2}\sqrt{\Big(\big(-1 - e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_2\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma^2\,\epsilon^2\,\eta_1\,\eta_2\,\xi_1\,\xi_2\big)^2 + 4\,e^{-\alpha0\,\gamma+\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\big(-e^{\gamma\,\alpha_1}\eta_1\,\xi_1 - \eta_2\,\xi_1 - e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1^2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\eta_2\,\xi_2 - 2\,e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_2^2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma^2\,\epsilon^2\,\eta_1\,\eta_2^2\,\xi_1^2\,\xi_2\big)\Big)}\Bigg)\Bigg)\Big/$$

$$\Big(\gamma\,\epsilon\,\big(\xi_1 + e^{\gamma\,\alpha_2}\xi_2 + e^{\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_2\,\xi_1\,\xi_2\big)\Big),\ \xi0 \to \Bigg(e^{-\gamma\,\alpha_2}\Bigg(\frac{1}{2\,\gamma\,\epsilon} + \frac{1}{2}e^{\gamma\,\alpha_1}\eta_1\,\xi_1 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\eta_1\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_2}\eta_2\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\,\xi_2 - \frac{1}{2\,\gamma\,\epsilon}\Big(\sqrt{\Big(\big(-1 - e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_2\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma^2\,\epsilon^2\,\eta_1\,\eta_2\,\xi_1\,\xi_2\big)^2 + 4\,e^{-\alpha0\,\gamma+\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\big(-e^{\gamma\,\alpha_1}\eta_1\,\xi_1 - \eta_2\,\xi_1 - e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1^2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\eta_2\,\xi_2 - 2\,e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_2}\gamma\,\epsilon\,\eta_2^2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\gamma^2\,\epsilon^2\,\eta_1\,\eta_2^2\,\xi_1^2\,\xi_2\big)\Big)}\Big)\Bigg)\Bigg)\Big/\ \big(e^{\gamma\,\alpha_1}\eta_1 + \eta_2 + e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\big)\Big\}$$

```
eqn = MatrixExp[η1 Cρ[CU@y]].MatrixExp[α1 Cρ[CU@a]].MatrixExp[ξ1 Cρ[CU@x]].
    MatrixExp[η2 Cρ[CU@y]].MatrixExp[α2 Cρ[CU@a]].MatrixExp[ξ2 Cρ[CU@x]] ==
  T0 MatrixExp[η0 Cρ[CU@y]].MatrixExp[α0 Cρ[CU@a]].MatrixExp[ξ0 Cρ[CU@x]]
```

$$\Big\{\Big\{e^{\gamma\,\alpha_2}\big(e^{\gamma\,\alpha_1} + e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_2\,\xi_1\big),\ e^{\gamma\,\alpha_1}\gamma\,\xi_1 + e^{\gamma\,\alpha_2}\gamma\,\big(e^{\gamma\,\alpha_1} + e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_2\,\xi_1\big)\,\xi_2\Big\},$$
$$\Big\{e^{\gamma\,\alpha_2}\big(e^{\gamma\,\alpha_1}\epsilon\,\eta_1 + \epsilon\,\eta_2\,\big(1 + e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1\big)\big),$$
$$1 + e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1 + e^{\gamma\,\alpha_2}\gamma\,\big(e^{\gamma\,\alpha_1}\epsilon\,\eta_1 + \epsilon\,\eta_2\,\big(1 + e^{\gamma\,\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1\big)\big)\,\xi_2\Big\}\Big\} ==$$
$$\Big\{\Big\{e^{\alpha0\,\gamma}\,T0,\ e^{\alpha0\,\gamma}\,T0\,\gamma\,\xi0\Big\},\ \Big\{e^{\alpha0\,\gamma}\,T0\,\epsilon\,\eta0,\ T0\,\big(1 + e^{\alpha0\,\gamma}\gamma\,\epsilon\,\eta0\,\xi0\big)\Big\}\Big\}$$

```
sol = Block[{ϵ}, Solve[Thread[Flatten /@ eqn], {T0, η0, α0, ξ0}]][[1]]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

$$\Big\{T0 \to \frac{1}{1 + \gamma\,\epsilon\,\eta_2\,\xi_1},\ \eta0 \to \frac{\eta_1 + e^{-\gamma\,\alpha_1}\eta_2 + \gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1}{1 + \gamma\,\epsilon\,\eta_2\,\xi_1},$$
$$\alpha0 \to \frac{\mathrm{Log}\big[e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\big(1 + \gamma\,\epsilon\,\eta_2\,\xi_1\big)^2\big]}{\gamma},\ \xi0 \to \frac{e^{-\gamma\,\alpha_2}\xi_1 + \xi_2 + \gamma\,\epsilon\,\eta_2\,\xi_1\,\xi_2}{1 + \gamma\,\epsilon\,\eta_2\,\xi_1}\Big\}$$

$\mathbb{E}$

$\mathbb{E}[\omega, L, Q, P]$ means $\omega e^{\hbar(L+Q)} P$, where ω is a scalar, L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial. It should be interpreted via $\mathrm{C}\mathbb{O}[\mathbb{E}[\,...],\ \{x_1,\ a_1,\ y_1\}_i,\ ...]$ (with some default for direct interpretation), or likewise via $\mathrm{Q}\mathbb{O}[\mathbb{E}[\,...],\ \{x_1,\ a_1,\ y_1\}_i,\ ...]$. In themselves, $\mathrm{C}\mathbb{O}$ and $\mathrm{Q}\mathbb{O}$ should have an interpretation in CU/QU by casting.